

Regenerierungszeit eines geothermischen Wärmereservoirs II

Falls der Wärmeleitungskoeffizient λ nicht konstant, sondern eine Funktion des Ortes ist, lässt sich die Wärmeleitungsgleichung

$$\frac{\partial T}{\partial t} = \frac{1}{\rho c} \operatorname{div}(\lambda \operatorname{grad} T) \quad (1)$$

im allgemeinen Fall nicht mehr analytisch lösen.

Der Wärmeleitungskoeffizient sei eine Funktion von r allein, d.h. $\lambda = \lambda(r)$, und das Gestein wird immer noch als isotrop und homogen vorausgesetzt. Die Lösung ist daher kugelsymmetrisch.

Zur numerischen Integration der Wärmeleitungsgleichung werden Kugelschalen mit Radius r und Dicke dr betrachtet.

Zur Zeit $t = 0$ ist die Temperatur im Innern der Kugel mit Radius R um ΔT niedriger als die Umgebungstemperatur. Während des ganzen Temperatúrausgleichs ist daher die Temperatur stets eine mit r monoton wachsende Funktion.

Die Kugelschale i mit dem inneren Radius r_i gibt nach innen an die Kugelschale $i - 1$ den Wärmestrom Q_1 ab und erhält von der Kugelschale $i + 1$ den Wärmestrom Q_2 . Die Wärmestromdichte ist

$$q = \lambda \frac{dT}{dr} \quad (2)$$

und für die Wärmeströme Q_1 und Q_2 ergibt sich

$$Q_1 = 4 \pi r_1^2 \lambda(r_1) \frac{T_i - T_{i-1}}{dr} \quad (3)$$

und

$$Q_2 = 4 \pi r_2^2 \lambda(r_2) \frac{T_{i+1} - T_i}{dr}. \quad (4)$$

Die Wärmezufuhr zur Kugelschale i während des Zeitintervalls dt ist

$$dQ = (Q_2 - Q_1) dt. \quad (5)$$

Durch diese Wärmezufuhr erwärmt sich die Kugelschicht i um dT_i :

$$dQ = 4 \pi r_m^2 dr \rho c dT_i. \quad (6)$$

$r_m = (r_i + r_{i+1})/2$ ist der mittlere Radius der Kugelschale i .

Die Temperaturänderung dT_i der Kugelschale i im Zeitintervall dt wird somit:

$$dT_i = \frac{Q_2 - Q_1}{4 \pi r_m^2 \rho c dr} dt. \quad (7)$$

Es gibt einen sehr einfachen Test für die Genauigkeit der numerischen Integration. Zur Zeit $t = 0$ wird der Temperaturverlauf beschrieben durch

$$T(r, 0) = \Delta T [\Theta(r - R) - 1], \quad (8)$$



Abbildung 1: Temperaturverlauf zur Zeit $t = 0$

und die dem Kugelvolumen $r < R$ entnommene Energie ist

$$E_0 = \int_0^R \rho c \Delta T 4 \pi r^2 dr. \quad (9)$$

Zur Zeit $t > 0$ ist die „fehlende“ Energie gegeben durch

$$E(t) = - \int_0^{r_\infty} \rho c T(r, t) 4 \pi r^2 dr. \quad (10)$$




Abbildung 2: Temperaturverlauf zur Zeit t

Die Integration ist bis zu einem Radius r_∞ zu erstrecken, für den die Temperatur(differenz) $T(r_\infty, t)$ praktisch gleich null wird.

Da der Temperatenausgleich nur innerhalb der Gesteinsschicht stattfindet und der Gesteinsschicht weder Energie entnommen noch zugeführt wird, muss auf Grund des Energieerhaltungssatzes für alle Zeiten $t > 0$ gelten:

$$E(t) = E_0. \quad (11)$$

Es zeigt sich, dass die analytische Lösung für $\lambda = \textit{konstant}$ diese Bedingung exakt erfüllt.

Die Wärmeleitfähigkeit von Wasser beträgt $0.6 \text{ Wm}^{-1}\text{K}^{-1}$, während Gestein eine Wärmeleitfähigkeit zwischen 2 und $4 \text{ Wm}^{-1}\text{K}^{-1}$ hat. Im Inneren der Kugel mit Radius R enthält das Gestein wassergefüllte Klüfte. Falls keine Konvektion auftritt, dürfte daher λ_1 im Inneren der Kugel kleiner sein als λ_2 ausserhalb. Bei starker Konvektion könnte auch $\lambda_1 > \lambda_2$ sein.

Wird für $\lambda(r)$ eine Sprungfunktion angenommen, d.h.

$$\lambda = \lambda_1 \text{ für } r < R \quad \text{und} \quad \lambda = \lambda_2 \text{ für } r > R, \quad (12)$$

so liefert die numerische Integration relativ grosse Abweichungen von der Bedingung 11. Daher wird ein stetiger Übergang von λ_1 zu λ_2 in Form einer Fermifunktion angenommen:

$$\lambda(r) = (\lambda_1 - \lambda_2) \frac{1}{e^{(r-R)/\Delta} + 1} + \lambda_2. \quad (13)$$

Für die Dicke Δ der Übergangszone kann z.B. 10 m eingesetzt werden.




Abbildung 3: $\lambda(r)$, $\lambda_1 = 5.0 \text{ Wm}^{-1}\text{K}^{-1}$ $\lambda_2 = 3.0 \text{ Wm}^{-1}\text{K}^{-1}$, $R = 500 \text{ m}$, $\Delta = 10 \text{ m}$




Abbildung 4: $\lambda(r)$, $\lambda_1 = 2.0 \text{ Wm}^{-1}\text{K}^{-1}$ $\lambda_2 = 3.0 \text{ Wm}^{-1}\text{K}^{-1}$, $R = 500 \text{ m}$, $\Delta = 10 \text{ m}$

Mit dieser Funktion $\lambda(r)$ ergeben sich bei der numerischen Integration Abweichungen von der Bedingung 11 im Bereich von etwa 1 Prozent.

In den folgenden Abbildungen stellt die blaue Kurve die analytische Lösung für konstante Wärmeleitfähigkeit $\lambda = 3.0 \text{ W}^{-1}\text{K}^{-1}$ dar. Die fuchsinrote Kurve ist die numerische Lösung für $\lambda(r)$.




Abbildung 5: $\lambda_1 = 5.0 \text{ Wm}^{-1}\text{K}^{-1}$, $\lambda_2 = 3.0 \text{ Wm}^{-1}\text{K}^{-1}$, Temperaturverlauf nach 1000 Jahren




Abbildung 6: $\lambda_1 = 2.0 \text{ Wm}^{-1}\text{K}^{-1}$, $\lambda_2 = 3.0 \text{ Wm}^{-1}\text{K}^{-1}$, Temperaturverlauf nach 1000 Jahren

Anhang: Pascal-Programm

```

Program geotherm;
{Das Programm berechnet die Zeit für die
Regenerierung eines geothermischen Reservoirs.}
{12.10., 13.10. 2011, A. Ruh}
{Numerische Integration der Differentialgleichung
mit ortsabhängiger Wärmeleitfähigkeit.}
{28.10., 29.10., 30.10.,
 4.11. 2011}
{Stetige Aenderung der Wärmeleitfähigkeit.}
{5.11. 2011}

{$N+}

uses crt, dos, graph, grphdriv, grphfnt0;

const
  rho = 2700.0;    {Dichte des Gesteins}
  c   = 800.0;    {Spezifische Wärmekapazität des Gesteins}
  la1 = 5.0;      {Wärmeleitfähigkeit für r<R}
  la2 = 3.0;      {Wärmeleitfähigkeit für r>R}
  d   = 5.0;      {Dicke der Uebergangszone der Fermifunktion}
  dt  = 20.0;     {Temperaturdifferenz Delta T}
  f   = 100;      {Zahl der Zeitschritte pro Jahr}
  L   = 5.0e3;    {Periodizitätsintervall}
  r0  = 500.0;    {Kugelradius R}
  ys  = 3.154e7;  {Anzahl Sekunden pro Jahr}
  sx  = 3.0;      {Skalenfaktor in x-Richtung}
  sy  = 5.0;      {Skalenfaktor in y-Richtung}
  x0  = 30;       {x-Koordinate des Nullpunkts}
  y0  = 100;      {y-Koordinate des Nullpunkts}
  zj  = 10;       {Zeit in Jahren}

type
  vector = array[1..600] of extended;

var
  gd : integer;
  gm : integer;
  i  : longint;   {Zähler der Zeitschritte}
  j  : integer;   {Zähler der Schritte in r-Richtung}
  j0 : integer;   {Nummer von r0}
  j1 : integer;   {j0+1}
  k  : integer;   {Index der Fourierterme}
  xi : integer;   {Bildschirmkoordinate}
  yi : integer;   {Bildschirmkordinate}
  zn : longint;   {Zahl der Zeitschritte}
  a  : extended;  {lambda/(rho*c)}
  b  : extended;  {Koeffizient der Fourierreihe}

```

```

fr : extended;    {Zahl der Zeitschritte pro Jahr}
jr : extended;    {r0/sx}
dr : extended;    {Dicke einer Kugelschale}
dz : extended;    {Zeitschritt}
lam: extended;    {Wärmeleitfähigkeit}
na : extended;    {Energie im Aeusseren der Kugel, analytisch}
ni : extended;    {Energie im Inneren der Kugel, analytisch}
nt : extended;    {totale Energie, analytisch}
q1 : extended;    {Wärmestrom von j nach j-1}
q2 : extended;    {Wärmestrom von j+1 nach j}
r  : extended;    {Radius}
r1 : extended;    {Innerer Radius der Kugelschale}
r2 : extended;    {Aeusserer Radius der Kugelschale}
rm : extended;    {Mittlerer Radius der Kugelschale}
s  : extended;    {Fouriersumme}
t  : extended;    {Zeit}
t0 : extended;    {Temperatur für r->0, analytisch}
tr : extended;    {T(r)}
tu : extended;    {Temperatur für r->0, numerisch}
va : extended;    {Energie im Aeusseren der Kugel, numerisch}
vi : extended;    {Energie im Inneren der Kugel, numerisch}
vt : extended;    {totale Energie, numerisch}
y  : extended;    {y-Koordinate}
z  : extended;    {Zeit}
u  : vector;      {Temperatur}
du : vector;      {Temperaturänderung}

```

```

begin
gd:=detect;
initgraph(gd,gm,'\tp7\bgi');
clearviewport;
bar(0,0,639,479);
setfillstyle(1,15);
setcolor(7);
line(x0,y0,630,y0);
line(x0,0,x0,250);
setcolor(10);
xi:=x0+round(r0/sx);
yi:=y0+round(sy*dt);
line(x0,yi,xi,yi);
line(xi,y0,xi,yi);
line(xi,y0,630,y0);
setcolor(9);
zn:=f*zj;
lam:=la2;
a:=lam/(rho*c);
dr:=1*sx;
va:=0.0;
vi:=0.0;
for i:=1 to 1 do

```

```

begin
  t:=i*ys*zj;
  for j:=1 to 600 do
  begin
    r:=sx*j;
    s:=0.0;
    for k:=1 to 100 do
    begin
      b:=-4*dt*(-r0/(2*pi*k)*cos(2*k*pi*r0/L)
        +L/(4*pi*pi*k*k)*sin(2*k*pi*r0/L));
      s:=s+exp(-a*sqr(2*k*pi/L)*t)*b/r*sin(2*k*pi*r/L);
    end;
    if r<r0 then
      vi:=vi+4*r*r*s*dr
    else
      va:=va+4*r*r*s*dr;
      vt:=(vi+va)*pi*rho*c;
      tr:=sy*s;
      yi:=y0-round(tr);
      xi:=x0+j;
      if j=1 then
      begin
        moveto(x0+1,yi);
        t0:=-100*s/dt;
      end
      else
        lineto(xi,yi);
      end;
    end;
  setcolor(0);
  outtextxy(15,15,'T');
  outtextxy(620,y0-15,'r');
  xi:=x0+round(r0/sx)-5;
  outtextxy(xi,y0-15,'R');
  readkey;

  jr:=r0/sx;
  j0:=round(jr);
  j1:=j0+1;
  z:=0.0;
  fr:=1.0*f;
  dz:=3.154e7/fr;
  na:=0.0;
  ni:=0.0;
  for j:=1 to j0 do
    u[j]:=-dt;
  for j:=j1 to 600 do
    u[j]:=0.0;
  for i:=1 to zn do
  begin

```

```

for j:=2 to 599 do
begin
  r1:=sx*j;
  r2:=sx*(j+1);
  rm:=0.5*(r1+r2);
  lam:=(la1-la2)/(exp((r1-r0)/d)+1.0)+la2;
  q1:=r1*r1*lam*(u[j]-u[j-1])/dr;
  lam:=(la1-la2)/(exp((r2-r0)/d)+1.0)+la2;
  q2:=r2*r2*lam*(u[j+1]-u[j])/dr;
  du[j]:=(q2-q1)*dz/(rm*rm*c*rho*dr);
end;
du[1]:=la1*(u[2]-u[1])*dz/(c*rho*dr*dr);
du[600]:=-la2*(u[600]-u[599])*dz/(c*rho*dr*dr);
for j:=1 to 600 do
  u[j]:=u[j]+du[j];
  z:=z+dz;
end;
for j:=1 to 600 do
begin
  r:=sx*j;
  if j<j0 then
    ni:=ni+4*r*r*u[j]*dr
  else
    if j>j0 then
      na:=na+4*r*r*u[j]*dr;
end;
nt:=(ni+na)*pi*rho*c;
for j:=1 to 600 do
begin
  y:=sy*u[j];
  yi:=y0-round(y);
  xi:=x0+j;
  setcolor(5);
  if j=1 then
    begin
      moveto(x0+1,yi);
      tu:=-100*u[j]/dt;
    end
  else
    lineto(xi,yi);
end;
end;
readkey;
closegraph;
restorecrtmode;
writeln('t0/Delta t = ',t0:5:1,' %');
writeln;
writeln('t0/Delta t = ',tu:5:1,' %');
writeln;
writeln('vi = ',vi:12);
writeln('va = ',va:12);

```



```
writeln('vt = ',vt:12);  
writeln;  
writeln('ni = ',ni:12);  
writeln('na = ',na:12);  
writeln('nt = ',nt:12);  
readkey;  
end.
```